

# VHISC Hardware Description Language


## Course Embedded Systems (SEm)

### VHDL




Silvan Zahno / François Corthay

Degree program Systems Engineering  
Specialization Infotronics – Embedded Systems



# Current content of the topic in the course



# VHISC Hardware Description Language (VHDL)

- Usage
- Structural description
- Behavioural description
- Usual types and operators
- Generic parameters
- Signals and variables
- Simulation and test

# VHDL

## Usage

### Basic usage

- Documentation
- Simulation
- Design
- Specification


### Extended usage

- Automated synthesis
- Delay verification
- Formal equivalence

# VHDL

## Caveat

Hardware description ! = Software program




# VHISC Hardware Description Language (VHDL)

- Usage
- **Structural description**
- Behavioural description
- Usual types and operators
- Generic parameters
- Signals and variables
- Simulation and test

# Structural Description

## Debouncing circuit

Circuit must remove bouncing for a given time period:



Will be used as an example throughout the presentation

# Structural Design


## Entity

The entity describes the external view of the circuit

```

1 library ieee;
2   use ieee.std_logic_1164.all;
3
4 entity debouncer is
5   port(
6     reset  : in  std_ulogic;
7     clock  : in  std_ulogic;
8     sigIn  : in  std_ulogic;
9     sigOut : out std_ulogic
10  );
11 end debouncer;
12

```






# Structural Description

## Debouncing design

Requires:

- A counter
- A Schmitt-trigger




# Structural Description

## Debouncing Architecture



A behavioral architecture describes components and signals

```

1 architecture struct of debouncer is
2
3   component upDownCounter
4     port(
5       reset    : in  std_ulogic;
6       clock    : in  std_ulogic;
7       up_down  : in  std_ulogic;
8       countOut : out unsigned(3 downto 0)
9     );
10  end component;
11
12  signal ramp : unsigned(3 downto 0);
13
14  component schmittTrigger
15    port(
16      valueIn : in  unsigned(3 downto 0);
17      trigOut : out std_ulogic
18    );
19  end component;
20
21 begin
22   -- components and connections
23 end struct;
24
  
```



upDownCounter



schmittTrigger


# Structural Description

## Debouncing Architecture

A behavioural architecture describes components and signals

```

1 architecture struct of debouncer is
2   -- signal and component declarations
3 begin
4   I_1 : upDownCounter
5   port map(
6     reset    => reset;
7     clock    => clock;
8     up_down  => sigIn;
9     countOut => ramp;
10  );
11  I_2 : schmittTrigger
12  port map(
13    valueIn => ramp;
14    trigOut => sigOut;
15  );
16 end struct;
  
```



# VHISC Hardware Description Language (VHDL)

- Usage
- Structural description
- **Behavioural description**
- Usual types and operators
- Generic parameters
- Signals and variables
- Simulation and test

# Behavioural Description

## Assignments



```

1 architecture RTL of schmittTrigger is
2   signal D, Q : std_ulogic;           -- flipflop I/O
3   signal gtHigh, ltLow : std_ulogic;   -- comparators
4   signal DSet : std_ulogic;           -- when over high threshold
5 begin
6   -- comparisons
7
8   DSet <= Q or gtHigh;                 -- gtHigh forces '1'
9   D <= DSet and not ltLow;            -- ltLow forces '0'
10
11   -- D-flipflop
12 end RTL;
13

```



```

1 D <= DSet and not ltLow;             -- ltLow forces '0'
2 DSet <= Q or gtHigh;                 -- gtHigh forces '1'

```

Whatever the order, the functionality is identical!

# Behavioural Description

## Processes

```

1 architecture RTL of schmittTrigger is
2   -- signal declarations
3 begin
4   -- comparisons and flipflop
5
6   p1: process(Q, gtHigh)
7   begin
8     DSet <= Q or gtHigh;           -- gtHigh forces '1'
9   end process p1;
10
11  p2: process(DSet, ltLow)
12  begin
13    D <= DSet and not ltLow;       -- ltLow forces '0'
14  end process p2;
15
16 end RTL;

```

- Direct assignments are a shorter way to declare what actually are processes
- Signals in the sensitivity list wake-up processes

# Behavioural Description

## Sensitivity list

```

1 architecture RTL of schmittTrigger is
2   -- signal declarations
3 begin
4   -- comparisons and flipflop
5
6   p1: process(Q, gtHigh, ltLow, DSet)
7   begin
8     D <= DSet and not ltLow;           -- ltLow forces '0'
9     DSet <= Q or gtHigh;               -- gtHigh forces '1'
10  end process p1;
11
12 end RTL;

```

- Several assignments can be written inside a process
- The assignments are done in parallel : one can consider that they effectively occur on process exit
- For combinatorial circuits, all signals on the right side of assignments have to be in the sensitivity list

# Behavioural Description

## Delays

```
1 architecture RTL of schmittTrigger is
2   -- signal declarations
3 begin
4   -- comparisons and flipflop
5
6   DSet <= Q or gtHigh after 2 ns;
7   D <= DSet and not ltLow after 1 ns;
8
9 end RTL;
```

- Assigned signals change a specified time after the process has been waken-up
- The simulator keeps a list of what is going to happen in the future
- The list can be overwritten by another assignment



# Behavioural Description

## Delay types

```
1 DSet <= Q or gtHigh after 2 ns;  
2  
3 DSet <= Q or gtHigh inertial after 2 ns;  
4  
5 DSet <= Q or gtHigh transport 2 ns;  
6  
7 DSet <= Q or gtHigh reject 1 ns after 2 ns;
```

- Different behaviours can be specified for impulses shorter than the delay
- Default is : i n e r t i a l

# Behavioural Description

## Conditions i f - e l s e

```

1 architecture RTL of schmittTrigger is
2   signal D, Q : std_ulogic;
3   signal gtHigh, ltLow : std_ulogic;
4 begin
5   -- comparisons and flipflop
6
7   setOrReset: process(gtHigh, ltLow, Q)
8   begin
9     if gtHigh = '1' then
10      D <= '1';
11    elsif ltLow = '1' then
12      D <= '0';
13    else
14      D <= Q;
15    end if;
16  end process setOrReset;
17
18 end RTL;

```

-- flipflop I/O  
-- comparators

- i f only within a process, condition in sensitivity list
- Combinatorial => e l s e is necessary

# Behavioural Description

## Conditions when

```

1 architecture RTL of schmittTrigger is
2   signal D, Q : std_ulogic;                                -- flipflop I/O
3   signal gtHigh, ltLow : std_ulogic;                       -- comparators
4 begin
5   -- comparisons and flipflop
6
7   D <=  '1' when gtHigh = '1'
8       else '0' when ltLow = '1'
9       else Q;
10
11 end RTL;

```

- when only outside a process
- Combinatorial => el se is necessary
- Compact but not very legible

# Behavioural Description

## Conditions case

```

1 architecture RTL of schmittTrigger is
2   signal D, Q : std_ulogic;                                -- flipflop I/O
3   signal gtHigh, ltLow : std_ulogic;                        -- comparators
4   signal conditions : std_ulogic_vector(1 to 2);
5 begin
6   -- comparisons and flipflop
7
8   conditions <= (gtHigh, ltLow);
9
10  setOrReset: process(conditions, Q)
11  begin
12    case conditions is
13      when "00"    => D <= Q ;
14      when "10"    => D <= '1';
15      when "01"    => D <= '0';
16      when others => D <= '-';
17    end case;
18  end process setOrReset;
19
20 end RTL;

```

- case only inside a process, selector in sensitivity list
- Combinatorial => when others is necessary

# Behavioural Description

## Conditions with select

```

1 architecture RTL of schmittTrigger is
2   signal D, Q : std_ulogic;
3   signal gtHigh, ltLow : std_ulogic;
4   signal conditions : std_ulogic_vector(1 to 2);
5 begin
6   -- comparisons and flipflop
7
8   conditions <= (gtHigh, ltLow);
9
10  with conditions select
11    D <=
12      Q when "00",
13      '1' when "10",
14      '0' when "01",
15      '-' when others;
16
17 end RTL;

```

-- flipflop I/O  
 -- comparators

- with only outside a process
- Combinatorial => when others is necessary
- Compact but less easy to read

# Behavioural Description

## Synchronous circuit

```

1 architecture RTL of schmittTrigger is
2   signal D, Q : std_ulogic;           -- flipflop I/O
3 begin
4   -- comparisons
5   -- calculation of D
6
7   store: process(reset, clock)
8   begin
9     if reset = '1' then
10      Q <= '0';
11    elsif rising_edge(clock) then
12      Q <= D;
13    end if;
14  end process store;
15
16 end RTL;

```

- All signals assigned within `rising_edge(clock)` are to be initialized at reset (and no other ones)

# Behavioural Description

## Sensitivity list

- Combinatorial circuits : all inputs
- Sequential circuits : only reset and clock

# VHISC Hardware Description Language (VHDL)

- Usage
- Structural description
- Behavioural description
- **Usual types and operators**
- Generic parameters
- Signals and variables
- Simulation and test



# Usual types and operators

## Numbers

```

1 type    integer is range -2147483648 to 2147483647;
2 subtype natural is integer range 0 to integer'high;
3 subtype positive is integer range 1 to integer'high;
4
5 -- operators:  + - * /    **    abs    mod rem    sla sra
6
7 -- comparisons:  =    /=    <    >    <=    >=
8

```

- Independent of their possible binary coding
- Not compatible with bit (or std\_logic) vectors

# Usual types and operators

## Logic signals

```

1 type std_ulogic is
2   (
3     'U', -- uninitialized
4     'X', -- forcing unknown
5     '0', -- forcing 0
6     '1', -- forcing 1
7     'Z', -- high impedance
8     'W', -- weak unknown
9     'L', -- weak 0
10    'H', -- weak 1
11    '-', -- don't care
12  );
13
14 -- operators:    not    and    nand    or    nor    xor
15
16 -- functions:    rising_edge    falling_edge
17

```

- Defined in ieee package std\_logic\_1164
- Signals always start with leftmost value : here ' U'

# Usual types and operators

## Logic signals

```

1 type std_ulogic is
2   (
3     'U', -- uninitialized
4     'X', -- forcing unknown
5     '0', -- forcing 0
6     '1', -- forcing 1
7     'Z', -- high impedance
8     'W', -- weak unknown
9     'L', -- weak 0
10    'H', -- weak 1
11    '-', -- don't care
12  );
13
14 -- operators:    not    and    nand    or    nor    xor
15
16 -- functions:    rising_edge    falling_edge
17

```

- Defined in ieee package std\_logic\_1164
- Signals always start with leftmost value : here ' U'

# Usual types and operators

## Logic arrays



```
1 type std_ulogic_vector is array ( natural range <> ) of std_ulogic;
2
3 -- concatenation operator for arrays:    &
```



```
1 type unsigned is array ( natural range <> ) of std_logic;
2 type signed   is array ( natural range <> ) of std_logic;
```



```
1 -- array assignments:
2 myString <= "food";
3 myString <= ('f', 'o', 'o', 'd');
4 myString <= (1 => 'f', 4 => 'd', 2|3 => 'o');
5 myString <= (4 => 'd', 1 => 'f', others => 'o');
6 myString <= (others => 'X');
```

- Types defined in IEEE packages `std_logic_1164` and `numeric_std`
- Operators to be examined in the corresponding package declarations

# Usual types and operators

## Type casting

```

1 architecture RTL of upDownCounter is
2   signal counter : unsigned(3 downto 0);
3   constant maxValue : signed(3 downto 0) := to_signed(-1,
   countOut'length);
4 begin
5   countAndSaturate: process(reset, clock)
6   begin
7     if reset = '1' then
8       counter <= (others => '0');
9     elsif rising_edge(clock) then
10      if (up_down = '1') and (counter <= unsigned(maxValue)) then
11        counter <= counter + 1;
12      elsif (up_down = '0') and (counter >= 0) then
13        counter <= counter - 1;
14      end if;
15    end if;
16  end process countAndSaturate;
17  countOut <= counter;
18 end RTL;

```

- VHDL is very strict
- signed maxValue has to be cast to unsigned
- Types must be compatible (here: array of std\_logic)

# Usual types and operators

## Type qualification

```

1 architecture RTL of upDownCounter is
2   subtype counterType is unsigned(3 downto 0);
3   signal counter : counterType;
4 begin
5   countAndSaturate: process(reset, clock)
6   begin
7     if reset = '1' then
8       counter <= (others => '0');
9     elsif rising_edge(clock) then
10      if (up_down = '1') and (counter+1 /= 0) then
11        counter <= counter + 1;
12      elsif (up_down = '0') and (counter >= counterType('0'))
13      then
14        counter <= counter - 1;
15      end if;
16    end if;
17  end process countAndSaturate;
18  countOut <= counter;
19 end RTL;

```

- Indeed, VHDL is very strict
- ( o t h e r s => ' 0' ) is not qualified : it may be of various lengths and of type string, unsigned, signed, ...

# Usual types and operators

## Type conversion

```


1 architecture RTL of upDownCounter is
2   signal counter : unsigned(3 downto 0);
3 begin
4   countAndSaturate: process(reset, clock)
5   begin
6     if reset = '1' then
7       counter <= to_unsigned(0, 4);
8     elsif rising_edge(clock) then
9       if (up_down = '1') and (counter+1 /= 0) then
10        counter <= counter + 1;
11      elsif (up_down = '0') and (counter >= 0) then
12        counter <= counter - 1;
13      end if;
14    end if;
15  end process countAndSaturate;
16  countOut <= counter;
17 end RTL;

```

- counter is unsigned, 0 is integer (or natural) : the types are not compatible
- Conversion function found in package numeric\_std

# Usual types and operators


## Usual castings and conversions





# Usual types and operators

## Physical types



```

1 type time is range -2147483647 to 2147483647
2   units
3     fs;
4     ps = 1000 fs;
5     ns = 1000 ps;
6     us = 1000 ns;
7     ms = 1000 us;
8     sec = 1000 ms;
9     min = 60 sec;
10    hr = 60 min;
11  end units;
  
```

- Mostly used for the type `time`
- Could be used for analog and mechanical simulations and others...

# VHISC Hardware Description Language (VHDL)

- Usage
- Structural description
- Behavioural description
- Usual types and operators
- **Generic parameters**
- Signals and variables
- Simulation and test


# Generic parameters

## Generic declaration

```

1 library ieee;
2   use ieee.std_logic_1164.all;
3
4 entity upDownCounter is
5   generic(
6     bitNb : integer := 8
7   )
8   port(
9     reset    : in  std_ulogic;
10    clock     : in  std_ulogic;
11    up_down   : in  std_ulogic;
12    countOut  : out unsigned(bitNb-1 downto 0)
13  );
14 end upDownCounter;

```



- The generic value is known in the port mapping and inside the circuit

# Generic Parameters

## Attributes

```

1 architecture RTL of upDownCounter is
2   signal counter : unsigned(countOut'range);
3 begin
4   countAndSaturate: process(reset, clock)
5   begin
6     if reset = '1' then
7       counter <= (others => '0');
8     elsif rising_edge(clock) then
9       if up_down = '1' then
10        if (counter(counter'high downto counter'high-1) /= "11") then
11          counter <= counter + 1;
12        end if;
13      elsif up_down = '0' then
14        if (counter(counter'high downto counter'high-1) /= "00") then
15          counter <= counter - 1;
16        end if;
17      end if;
18    end if;
19  end process countAndSaturate;
20  countOut <= counter;
21 end RTL;

```

- Signal and type attributes allow to write robust code
- Often independent of generic parameters

# Generic Parameters

## Attributes

```

1 architecture RTL of upDownCounter is
2   signal counter : unsigned(7 downto 0);
3   signal debugString : string(1 to 10);
4 begin
5   -- counter'length = 8
6   -- counter'high   = 7
7   -- counter'low    = 0
8   -- counter'left   = 7
9   -- counter'right  = 0
10  -- counter'range   = (7 downto 0)
11  -- counter'reverse_range = (0 to 7)
12
13  -- counter MSB = counter(7) = counter(counter'high)
14
15  -- debugString'left   = 1
16  -- debugString'right  = 10
17 end RTL;
  
```

- Never use constants within the code : prefer the attributes
- The code must function with different generic values

# Generic Parameters

## Iterative structures

```

1 architecture structural of shiftRegister is
2
3   component dFlipFlop
4     port(
5       clock, reset, d : in std_ulogic;
6       Q               : out std_ulogic
7     );
8   end component;
9
10  signal c: std_ulogic_vector(1 to bitNb+1); -- 1 bit more
11
12  begin
13
14    c(c'low) <= dataIn;
15
16    shReg: for i in c'low to c'high-1 generate
17      dFF: dFlipFlop port map(clock, reset, c(i), c(i+1));
18    end generate shReg;
19
20    dataOut <= c(c'high);
21
22  end structural;

```

- Describe circuits with sizes based on generic values

# VHISC Hardware Description Language (VHDL)

- Usage
- Structural description
- Behavioural description
- Usual types and operators
- Generic parameters
- **Signals and variables**
- Simulation and test

# Signals and variables

```

1 architecture RTL of schmittTrigger is
2   signal DSet : std_ulogic
3   -- other signal declarations
4 begin
5   -- comparisons and flipflop
6
7   withSignal: process(Q, gtHigh, ltLow, DSet)
8   begin
9     D <= DSet and not ltLow;           -- ltLow
10    forces '0'
11    DSet <= Q or gtHigh;               -- gtHigh
12    forces '1'
13  end process withSignal;
14
15  withVariable: process(Q, gtHigh, ltLow)
16    variable DSet_var : std_ulogic;
17    begin
18      DSet_var := Q or gtHigh;         -- gtHigh
19      forces '1'
20      D <= DSet_var and not ltLow;     -- ltLow
21      forces '0'
22    end process withVariable;
23
24 end RTL;

```

- Signal s declared at architecture level
- Variables declared inside a process (and not seen outside it)



# Signals and variables

## Variables to signals

```

1 architecture RTL of upDownCounter is
2 begin
3   countAndSaturate: process(reset, clock)
4     variable counter : unsigned(countOut'range);
5     begin
6       if reset = '1' then
7         counter := (others = '0');
8       elsif rising_edge(clock) then
9         if up_down = '1' then
10          counter := counter + 1;
11          if counter = 0 then                                -- if
overflow
12            counter := (others = '1');                      -- back to
max value
13          end if;
14          elsif (up_down = '0') and (counter >= 0) then
15            counter := counter - 1;
16          end if;
17        end if;
18        countOut <= counter;                                -- inside the process, best at
the end
19      end process countAndSaturate;
20 end RTL;

```

- Like signals, variables keep their values between process activations

# Signals and variables

## Iterative functions

```

1 architecture RTL of upDownCounter is
2   signal counter : unsigned(countOut'range);
3   signal allZero : std_ulogic
4 begin
5
6   checkIfAllZero: process(counter)
7     variable allZeroInt : std_ulogic;
8     begin
9       allZeroInt := '1';
10      for i in counter'range loop
11        if counter(i) = '1' then
12          allZeroInt := '0';
13        end if;
14      end loop;
15      delay !
16      allZero <= allZeroInt;
17    end process checkIfAllZero;
18 end RTL;

```

-- loop with zero

- Here is a very good reason to use variables
- Loops are done over array elements, not in time: only the clock lets us advance in

# Signals and variables

## Usage of variables

- Variables allow programming-like code
- Remain unchanged between process activations
- Are fine for describing iterative structures
- Only known inside processes (not totally true)
- Cannot have a delay attached

# VHISC Hardware Description Language (VHDL)

- Usage
- Structural description
- Behavioural description
- Usual types and operators
- Generic parameters
- Signals and variables
- **Simulation and test**

# Simulation and test

## Simple testbench example

```

1 architecture test of upDownCounter_tester is
2                                     -- clock definitions
3     constant clockFrequency : real      := 100.0E6;
4     constant clockPeriod    : time      := (1.0/clockFrequency) * 1 sec;
5     signal sClock           : std_ulogic := '1';
6
7 begin
8     -----
9                                     -- clock and reset
10    sClock <= not sClock after clockPeriod/2;
11    clock <= transport sClock after clockPeriod*9/10;
12    reset <= '1', '0' after 2*clockPeriod;
13
14    -----
15                                     -- input signal waveform
16    up_down <=
17        '0',
18        '1' after 10*clockPeriod,
19        '0' after 50*clockPeriod;
20
21 end test;

```

- Weird types and assignments, not usable in synthesizable code

# Simulation and test

## Initial values

```

1 architecture RTL of upDownCounter is
2   signal counter : unsigned(countOut'range) := (others => '0');
3   signal isHigh : std_ulogic := '0';
4 begin
5
6   countAndSaturate: process(reset, clock)
7   begin
8     if reset = '1' then
9       counter <= (others => '0');
10    elsif rising_edge(clock) then
11      if (up_down = '1') and (counter+1 /= 0) then
12        counter <= counter + 1;
13      elsif (up_down = '0') and (counter >= 0) then
14        counter <= counter - 1;
15      end if;
16    end if;
17  end process countAndSaturate;
18
19  isHigh <= '1' when counter(counter'high downto counter'high-1) = "11"
20    else '0';
21
22 end RTL;

```

- Only flipflops are initialized, with the reset signal
- Combinatorial logic can't be initialized

# Simulation and test

## Memory effect (latch)



```
1 architecture RTL of schmittTrigger is
2   signal D, Q : std_ulogic;
3   signal gtHigh, ltLow : std_ulogic;
4 begin
5   -- comparisons and flipflop
6
7   setOrReset: process(gtHigh, ltLow, Q)
8   begin
9     if gtHigh = '1' then
10      D <= '1';
11    elsif ltLow = '1' then
12      D <= '0';
13    end if;
14  end process setOrReset;
15
16 end RTL;
```

- No else condition => D might not be assigned => has to keep the same value
- The synthesizer will try to implement a latch
- The circuit will not always work (maybe sometimes)

# Simulation and test

## Test vs synthesizable code

- Testbench is not going into a (programmable) circuit
- Some signal types and functions are not synthesizable
- Delays are not synthesizable
- Initializations are only synthesizable with the help of a Power-On Reset (POR)




# Simulation and test

## Real world

- Simulation ensures proper functionality as long as:
  - the stimuli correspond to the real world signals
  - the circuit is synchronous
  - the inputs are properly synchronized
  - the synthesis has not reported severe problems

# VHDL

- We now know everything about:
  - Structural description
  - Behavioural description
  - Usual types and operators
  - Generic parameters
  - Signals and variables
  - Simulation and test



# References

- [War17] (Englisch) FPGA Designer Warrior  
[http://blog.aku.edu.tr/ismailkoyuncu/files/2017/04/01\\_ebook.pdf](http://blog.aku.edu.tr/ismailkoyuncu/files/2017/04/01_ebook.pdf)
- [Int19] (Englisch) Intel FPGA Website  
<https://www.intel.com/content/www/us/en/products/programmable.html>
- [Xil19] (Englisch) Xilinx FPGA Website  
<https://www.xilinx.com/>
- [Act19] (Englisch) Actel FPGA Website  
<https://www.microsemi.com/product-directory/1636-fpga-soc>

